Chapter 22
Multi-ζ Functions

22.1 Definition

The Riemann ζ-function is defined by

$$\zeta_s = \sum_{n>0} \frac{1}{n^s}\,.$$

Mathematica knows this function; it can be expressed via powers of π for even integer values of s.

In[1] := Table[Zeta[*s*], {*s*, 2, 6}]

Out[1] = $\left\{\frac{\pi^2}{6}, \text{Zeta}[3], \frac{\pi^4}{90}, \text{Zeta}[5], \frac{\pi^6}{945}\right\}$

Let's define

$$\zeta_{s_1 s_2} = \sum_{n_1>n_2>0} \frac{1}{n_1^{s_1} n_2^{s_2}}\,, \qquad \zeta_{s_1 s_2 s_3} = \sum_{n_1>n_2>n_3>0} \frac{1}{n_1^{s_1} n_2^{s_2} n_3^{s_3}}\,,$$

and so on. These series converge at $s_1 > 1$. *Mathematica* does not know these multi-ζ functions. The sum $s_1 + s_2 + \cdots + s_k$ is called the *weight*. All relations we shall discuss contain terms of the same weight (the weight of a product is the sum of the weights of its factors).

22.2 Stuffling Relations

Suppose we want to multiply $\zeta_s \zeta_{s_1 s_2}$:

$$\zeta_s \zeta_{s_1 s_2} = \sum_{\substack{n>0 \\ n_1>n_2>0}} \frac{1}{n^s n_1^{s_1} n_2^{s_2}}\,.$$

A. Grozin, *Introduction to Mathematica® for Physicists*, Graduate Texts in Physics,
DOI 10.1007/978-3-319-00894-3_22,

Here n can be anywhere with respect to n_1, n_2. There are five contributions:

$$\sum_{n>n_1>n_2>0} \frac{1}{n^s n_1^{s_1} n_2^{s_2}} = \zeta_{ss_1s_2}\,,$$
$$\sum_{n=n_1>n_2>0} \frac{1}{n^s n_1^{s_1} n_2^{s_2}} = \zeta_{s+s_1,s_2}\,,$$
$$\sum_{n_1>n>n_2>0} \frac{1}{n^s n_1^{s_1} n_2^{s_2}} = \zeta_{s_1ss_2}\,,$$
$$\sum_{n_1>n=n_2>0} \frac{1}{n^s n_1^{s_1} n_2^{s_2}} = \zeta_{s_1,s+s_2}\,,$$
$$\sum_{n_1>n_2>n>0} \frac{1}{n^s n_1^{s_1} n_2^{s_2}} = \zeta_{s_1s_2s}\,.$$

This process reminds shuffling cards. The order of cards in the upper deck, as well as in the lower one, is kept fixed. We sum over all possible shufflings. Unlike real playing cards, however, two cards may be exactly on top of each other. In this case they are stuffed together: a single card (which is their sum) appears in the resulting deck. A mathematical jargon term for such shuffling with (possible) stuffing is *stuffling*.

In[2] := Show[Import["c1.jpg"]]

Out[2] =



Let's implement this in *Mathematica*. The multi-ζ function will be called ζ; it can have any number of arguments. The function Stuffling first of all transforms products of ζ functions (including squares) to a local function z with three list parameters: the first two contain the arguments of the initial ζ functions, and the third one is empty. These are our two decks for shuffling and the resulting deck, initially empty. All the work is done by the following repeated substitution. Let the two unprocessed decks be nonempty: the first one contains some front "card" $a\_$ and the remainder $A\_\_\_$; the second one—the front "card" $b\_$ and the remainder $B\_\_\_$. Then there are three possibilities: either we move the front "card" from the first deck (a) to the resulting deck, or we move the front "card" from the second deck (b) to the resulting deck, or we take the front "cards" from both decks and put their sum to the resulting deck (stuffing). We need to use a delayed substitution :> here to ensure that the command Expand in its right-hand side is executed when the substitution is applied. In addition to this, we should take care of the situations when one of the source decks becomes empty. In this case we can just append the other source deck to the resulting one and yield the result.

This process can also be described as the following. The final result is a sum of many ζ functions with various argument lists. During intermediate steps, the function z[deck1, deck2, res] represents the sum of a subset of terms of the result whose argument lists begin with res. At each step we subdivide this sum into three parts, according to three possible values of the next argument.

```
In[3] := Stuffling[x_] := Module[{y,z},y = x/.
    {ζ[A___]^2->z[{A},{A},{}],ζ[A___]*ζ[B___]->z[{A},{B},{}]};
  y//.{z[{},{B___},{C___}]->ζ[C,B],z[{A___},{},{C___}]->ζ[C,A],
    z[{a_,A___},{b_,B___},{C___}] :>
      Expand[z[{A},{b,B},{C,a}] + z[{a,A},{B},{C,b}]+
        z[{A},{B},{C,a+b}]]}]
In[4] := Map[Stuffling,{ζ[x]^2,ζ[x]*ζ[y],ζ[x]*ζ[y,z]}]
Out[4] = {ζ[2x] + 2ζ[x,x],ζ[x+y] + ζ[x,y] + ζ[y,x],
    ζ[y,x+z] + ζ[x+y,z] + ζ[x,y,z] + ζ[y,x,z] + ζ[y,z,x]}
```

22.3 Integral Representation

It is easy to check the integral representation of the ζ-function of an integer argument

$$\zeta_s = \int\limits_{1>x_1>\cdots>x_s>0} \frac{\mathrm{d}x_1}{x_1} \cdots \frac{\mathrm{d}x_{s-1}}{x_{s-1}} \frac{\mathrm{d}x_s}{1-x_s}.$$

Let's denote

$$\omega_0 = \frac{\mathrm{d}x}{x}, \qquad \omega_1 = \frac{\mathrm{d}x}{1-x}.$$

All integrals will always have the integration region $1 > x_1 > \cdots > x_s > 0$. Then

$$\zeta_s = \int \omega_0^{s-1} \omega_1 \,.$$

This representation can be generalized to multi-ζ functions:

$$\zeta_{s_1 s_2} = \int \omega_0^{s_1-1} \omega_1 \omega_0^{s_2-1} \omega_1 \,, \qquad \zeta_{s_1 s_2 s_3} = \int \omega_0^{s_1-1} \omega_1 \omega_0^{s_2-1} \omega_1 \omega_0^{s_3-1} \omega_1 \,,$$

and so on. Let's write functions for transforming ζ with integer arguments to such integral representation and back. The integral representation ζi takes an arbitrary number of arguments equal to 0 or 1 corresponding to ω_0, ω_1.

```
In[5] := s2i[x_] := Module[{y,z}, y = x/. ζ[A___]->z[{A},{}];
    y//. {z[{},{B___}]->ζi[B],
        z[{a_,A___},{B___}] :>
            z[{A}, Append[Join[{B}, Table[0,{a-1}]], 1]]}]
In[6] := l = Map[s2i, {ζ[2], ζ[2,3]}]
Out[6] = {ζi[0,1], ζi[0,1,0,0,1]}
In[7] := i2s[x_] := Module[{y,z}, y = x/. ζi[A___]->z[{A},{1}];
    y//. {z[{1},{B___}]->ζ[B],
        z[{a_,A___},{B___,b_}] :>
            If[a == 0, z[{A},{B,b+1}], z[{A},{B,b,1}]]}]
In[8] := Map[i2s, l]
Out[8] = {ζ[2], ζ[2,3]}
In[9] := Clear[l]
```

22.4 Shuffling Relations

Suppose we want to multiply $\zeta_2 \cdot \zeta_2$:

$$\zeta_2^2 = \int\limits_{1>x_1>x_2>0} \omega_0 \omega_1 \cdot \int\limits_{1>x_1'>x_2'>0} \omega_0 \omega_1 \,.$$

The order of primed and non-primed integration variables is not fixed. There are six contributions:

$$1 > x_1 > x_2 > x_1' > x_2' > 0 : \quad \int \omega_0 \omega_1 \omega_0 \omega_1 = \zeta_{22} \,;$$

$$1 > x_1 > x_1' > x_2 > x_2' > 0 : \quad \int \omega_0 \omega_0 \omega_1 \omega_1 = \zeta_{31} \,;$$

$$1 > x_1 > x_1' > x_2' > x_2 > 0 : \quad \int \omega_0 \omega_0 \omega_1 \omega_1 = \zeta_{31} \,;$$

$$1 > x_1' > x_1 > x_2 > x_2' > 0: \quad \int \omega_0\omega_0\omega_1\omega_1 = \zeta_{31};$$

$$1 > x_1' > x_1 > x_2' > x_2 > 0: \quad \int \omega_0\omega_0\omega_1\omega_1 = \zeta_{31};$$

$$1 > x_1' > x_2' > x_1 > x_2 > 0: \quad \int \omega_0\omega_1\omega_0\omega_1 = \zeta_{22}.$$

Now we are multiplying integrals, not sums. Therefore our "cards" are now infinitely thin and cannot be exactly on top of each other. There are just two kinds of "cards": ω_0 and ω_1, and we sum over all possible shufflings of two decks.

```
In[10] := Show[Import["c2.jpg"]]
```

Out[10] =



```
In[11] := shuffling[x_] := Module[{y,z},y = x/.
      {ζi[A___]^2->z[{A},{A},{}],
        ζi[A___]*ζi[B___]->z[{A},{B},{}]};
   y//.{z[{},{B___},{C___}]->ζi[C,B],z[{A___},{},{C___}]->ζi[C,A],
      z[{a_,A___},{b_,B___},{C___}] :>
        Expand[z[{A},{b,B},{C,a}]+z[{a,A},{B},{C,b}]]}]
In[12] := Shuffling[x_] := i2s[Expand[shuffling[s2i[x]]]]
In[13] := Map[Shuffling,{ζ[2]^2,ζ[2]*ζ[3],ζ[2]*ζ[2,1]}]
Out[13] = {2ζ[2,2]+4ζ[3,1],ζ[2,3]+3ζ[3,2]+6ζ[4,1],
      ζ[2,1,2]+3ζ[2,2,1]+6ζ[3,1,1]}
```

22.5 Duality Relations

The integral representation allows us to derive another set of useful relations, even simpler than shuffling—duality relations. Let's make the substitution $x_i \to 1 - x_i$. Then $\omega_0 \longleftrightarrow \omega_1$; to preserve the order $1 > x_1 > \cdots > x_s > 0$, we have to arrange all the ω factors in the opposite order. In other words, after writing down an integral representation for a multi-ζ value, we may read it in the Arabic fashion, right to left, simultaneously replacing $\omega_0 \longleftrightarrow \omega_1$. Duality relations are the only known relations which say that two multi-ζ values with distinct arguments are just equal to each other.

```
In[14] := duality[x_] := Module[{y,z},y = x/.ζi[A___]->z[{A},{}];
    y//.{z[{},{B___}]->ζi[B],z[{a_,A___},{B___}]->z[{A},{1-a,B}]}]
In[15] := Duality[x_] := i2s[duality[s2i[x]]]
In[16] := Map[Duality,{ζ[3],ζ[4],ζ[5],ζ[4,1],ζ[3,2],ζ[2,3]}]
Out[16] = {ζ[2,1],ζ[2,1,1],ζ[2,1,1,1],ζ[3,1,1],ζ[2,2,1],ζ[2,1,2]}
```

22.6 Weight 4

There are four converging multi-ζ series of weight 4: ζ_4, ζ_{31}, ζ_{22}, and ζ_{211}. Due to duality, two of them are equal to each other.

```
In[17] := Duality[ζ[4]]
Out[17] = ζ[2,1,1]
```

We can express ζ_4 via ζ_2^2 using their explicit values:

```
In[18] := S = ζ[4]->Zeta[4]/Zeta[2]^2*ζ[2]^2
```

$$\mathrm{Out}[18] = \zeta[4] \to \frac{2\zeta[2]^2}{5}$$

Two equations for ζ_2^2 follow from stuffling

```
In[19] := eq1 = ζ[2]^2 == Stuffling[ζ[2]^2]
```

$$\mathrm{Out}[19] = \zeta[2]^2 == \zeta[4] + 2\zeta[2,2]$$

and shuffling

```
In[20] := eq2 = ζ[2]^2 == Shuffling[ζ[2]^2]
```

$$\mathrm{Out}[20] = \zeta[2]^2 == 2\zeta[2,2] + 4\zeta[3,1]$$

They can be solved for ζ_{22} and ζ_{31} (taking the expression for ζ_4 into account).

```
In[21] := s = Solve[{eq1/.S,eq2},{ζ[2,2],ζ[3,1]}][[1]]
```

$$\mathrm{Out}[21] = \left\{\zeta[2,2] \to \frac{3\zeta[2]^2}{10}, \zeta[3,1] \to \frac{\zeta[2]^2}{10}\right\}$$

Thus we have demonstrated that all multi-ζ values of weight 4 can be expressed via ζ_2^2.

```
In[22] := Clear[S]
```

22.7 Weight 5

There are four distinct multi-ζ values of weight 5,

In[23] := Map[Duality, {ζ[5], ζ[4, 1], ζ[3, 2], ζ[2, 3]}]

Out[23] = {ζ[2, 1, 1, 1], ζ[3, 1, 1], ζ[2, 2, 1], ζ[2, 1, 2]}

due to duality. The stuffling relation for $\zeta_2\zeta_3$:

In[24] := eq1 = ζ[2] ∗ ζ[3] == Stuffling[ζ[2] ∗ ζ[3]]

Out[24] = ζ[2]ζ[3] == ζ[5] + ζ[2, 3] + ζ[3, 2]

A similar equation where ζ_3 is written in the dual form.

In[25] := eq2 = Stuffling[ζ[2] ∗ Duality[ζ[3]]]

Out[25] = ζ[2, 3] + ζ[4, 1] + ζ[2, 1, 2] + 2ζ[2, 2, 1]

In[26] := eq2 = eq2/. {ζ[2, 1, 2] : >Duality[ζ[2, 1, 2]],
ζ[2, 2, 1] : >Duality[ζ[2, 2, 1]]}

Out[26] = 2ζ[2, 3] + 2ζ[3, 2] + ζ[4, 1]

In[27] := eq2 = ζ[2] ∗ ζ[3] == eq2

Out[27] = ζ[2]ζ[3] == 2ζ[2, 3] + 2ζ[3, 2] + ζ[4, 1]

The shuffling relation for $\zeta_2\zeta_3$:

In[28] := eq3 = ζ[2] ∗ ζ[3] == Shuffling[ζ[2] ∗ ζ[3]]

Out[28] = ζ[2]ζ[3] == ζ[2, 3] + 3ζ[3, 2] + 6ζ[4, 1]

This system can be solved for ζ_{41}, ζ_{32}, and ζ_{23}.

In[29] := s = Solve[{eq1, eq2, eq3}, {ζ[4, 1], ζ[3, 2], ζ[2, 3]}][[1]]

Out[29] = $\left\{\zeta[4,1] \to -\zeta[2]\zeta[3] + 2\zeta[5], \zeta[3,2] \to 3\zeta[2]\zeta[3] - \frac{11\zeta[5]}{2},\right.$
$\left.\zeta[2,3] \to -2\zeta[2]\zeta[3] + \frac{9\zeta[5]}{2}\right\}$

Thus we have demonstrated that all multi-ζ values of weight 5 can be expressed via $\zeta_2\zeta_3$ and ζ_5.